

# NAG Library Routine Document

## G01ECF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

### 1 Purpose

G01ECF returns the lower or upper tail probability for the  $\chi^2$ -distribution with real degrees of freedom, via the routine name.

### 2 Specification

```
FUNCTION G01ECF (TAIL, X, DF, IFAIL)
REAL (KIND=nag_wp) G01ECF
INTEGER IFAIL
REAL (KIND=nag_wp) X, DF
CHARACTER(1) TAIL
```

### 3 Description

The lower tail probability for the  $\chi^2$ -distribution with  $\nu$  degrees of freedom,  $P(X \leq x : \nu)$  is defined by:

$$P(X \leq x : \nu) = \frac{1}{2^{\nu/2} \Gamma(\nu/2)} \int_{0.0}^x X^{\nu/2-1} e^{-X/2} dX, \quad x \geq 0, \nu > 0.$$

To calculate  $P(X \leq x : \nu)$  a transformation of a gamma distribution is employed, i.e., a  $\chi^2$ -distribution with  $\nu$  degrees of freedom is equal to a gamma distribution with scale parameter 2 and shape parameter  $\nu/2$ .

### 4 References

Abramowitz M and Stegun I A (1972) *Handbook of Mathematical Functions* (3rd Edition) Dover Publications

Hastings N A J and Peacock J B (1975) *Statistical Distributions* Butterworth

### 5 Parameters

- |  |              |
|--|--------------|
| 1: TAIL – CHARACTER(1)   | <i>Input</i> |
| <p><i>On entry:</i> indicates whether the upper or lower tail probability is required.</p>                                     |              |
| TAIL = 'L'   |              |
| <p>The lower tail probability is returned, i.e., <math>P(X \leq x : \nu)</math>.</p>   |              |
| TAIL = 'U'   |              |
| <p>The upper tail probability is returned, i.e., <math>P(X \geq x : \nu)</math>.</p>   |              |
| <p><i>Constraint:</i> TAIL = 'L' or 'U'.</p>   |              |
| 2: X – REAL (KIND=nag_wp)  | <i>Input</i> |
| <p><i>On entry:</i> <math>x</math>, the value of the <math>\chi^2</math> variate with <math>\nu</math> degrees of freedom.</p> |              |
| <p><i>Constraint:</i> <math>X \geq 0.0</math>.</p>   |              |

3: DF – REAL (KIND=nag\_wp) *Input*

*On entry:*  $\nu$ , the degrees of freedom of the  $\chi^2$ -distribution.

*Constraint:*  $DF > 0.0$ .

4: IFAIL – INTEGER *Input/Output*

*On entry:* IFAIL must be set to 0,  $-1$  or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value  $-1$  or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if  $IFAIL \neq 0$  on exit, the recommended value is  $-1$ . **When the value  $-1$  or 1 is used it is essential to test the value of IFAIL on exit.**

*On exit:*  $IFAIL = 0$  unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry  $IFAIL = 0$  or  $-1$ , explanatory error messages are output on the current error message unit (as defined by X04AAF).

**Note:** G01ECF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

If  $IFAIL = 1, 2$  or  $3$  on exit, then G01ECF returns 0.0.

$IFAIL = 1$

On entry,  $TAIL \neq 'L'$  or  $'U'$ .

$IFAIL = 2$

On entry,  $X < 0.0$ .

$IFAIL = 3$

On entry,  $DF \leq 0.0$ .

$IFAIL = 4$

The solution has failed to converge while calculating the gamma variate. The result returned should represent an approximation to the solution.

## 7 Accuracy

A relative accuracy of five significant figures is obtained in most cases.

## 8 Further Comments

For higher accuracy the transformation described in Section 3 may be used with a direct call to S14BAF.

## 9 Example

Values from various  $\chi^2$ -distributions are read, the lower tail probabilities calculated, and all these values printed out, until the end of data is reached.

## 9.1 Program Text

```

Program g01ecfe

!      G01ECF Example Program Text

!      Mark 24 Release. NAG Copyright 2012.

!      .. Use Statements ..
Use nag_library, Only: g01ecf, nag_wp
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter :: nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp) :: df, prob, x
Integer :: ifail
Character (1) :: tail
!      .. Executable Statements ..
Write (nout,*) 'G01ECF Example Program Results'
Write (nout,*)

!      Skip heading in data file
Read (nin,*)

!      Display titles
Write (nout,*) ' TAIL      X      DF      Probability'
Write (nout,*)

d_lp: Do
    Read (nin,*,Iostat;ifail) tail, x, df
    If (ifail/=0) Then
        Exit d_lp
    End If

    !      Calculate probability
    ifail = -1
    prob = g01ecf(tail,x,df,ifail)
    If (ifail/=0) Then
        If (ifail/=4) Then
            Exit d_lp
        End If
    End If

    !      Display results
    Write (nout,99999) tail, x, df, prob
End Do d_lp

99999 Format (3X,A1,4X,F6.3,F8.1,7X,F7.4)
End Program g01ecfe

```

## 9.2 Program Data

```

G01ECF Example Program Data
'L' 8.26 20.0 : TAIL X DF
'L' 6.2   7.5
'L' 55.76 45.0

```

## 9.3 Program Results

G01ECF Example Program Results

TAIL	X	DF	Probability
L	8.260	20.0	0.0100
L	6.200	7.5	0.4279
L	55.760	45.0	0.8694